



SAS Publishing



Proc FORMS

The correct bibliographic citation for this manual is as follows: SAS Institute Inc. 2004. *Proc FORMS*. Cary, NC: SAS Institute Inc.

Proc FORMS

Copyright © 2004, SAS Institute Inc., Cary, NC, USA

All rights reserved. Produced in the United States of America. No part of this publication may be reproduced, stored in a retrieval system, or transmitted, in any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

U.S. Government Restricted Rights Notice. Use, duplication, or disclosure of this software and related documentation by the U.S. government is subject to the Agreement with SAS Institute and the restrictions set forth in FAR 52.227-19 Commercial Computer Software-Restricted Rights (June 1987).

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, February 2004


SAS Publishing provides a complete selection of books and electronic products to help customers use SAS software to its fullest potential. For more information about our e-books, e-learning products, CDs, and hard-copy books, visit the SAS Publishing Web site at support.sas.com/pubs or call 1-800-727-3228.

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

Contents

Chapter 1	△	The FORMS Procedure	1
Overview:		FORMS Procedure	1
Syntax:		FORMS Procedure	3
Concepts:		FORMS Procedure	9
Examples:		FORMS Procedure	11



CHAPTER

1

The FORMS Procedure

<i>Overview: FORMS Procedure</i>	1
<i>Syntax: FORMS Procedure</i>	3
<i>PROC FORMS Statement</i>	3
<i>BY Statement</i>	7
<i>FREQ Statement</i>	7
<i>LINE Statement</i>	8
<i>Concepts: FORMS Procedure</i>	9
<i>Form Layout</i>	9
<i>Modes of Operation</i>	10
<i>PROC FORMS Operation Modes</i>	10
<i>In Continuous Mode, PROC FORMS Always Writes to an External File</i>	10
<i>In Page Mode, PROC FORMS Can Write Either to an External File or to the Procedure Output File</i>	10
<i>Examples: FORMS Procedure</i>	11
<i>Example 1: Printing a Single Form Unit for Each Observation</i>	11
<i>Example 2: Printing Two Sets of Mailing Labels</i>	14
<i>Example 3: Writing Multiple Copies of a Label within a Single Set of Labels</i>	16

Overview: FORMS Procedure

The FORMS procedure produces labels for envelopes, mailing labels, external tape labels, file cards, and any other printer forms that have a regular pattern.

For each observation in the input SAS data set, PROC FORMS prints data in a rectangular block called a *form unit*. For example, a mailing label is a form unit.

Output 1.1 illustrates a simple mailing list produced with PROC FORMS. The statements that produce the output follow. The OBS= data set option limits to six the number of observations that PROC FORMS processes.

```
options pagesize=60 linesize=64 nodate
        pageno=1;

filename labels 'external-file';

proc forms data=list(obs=6) file=labels
        align=0;
    line 1 name;
    line 2 street;
    line 3 city state zip;
```

run;

Output 1.1 Simple Mailing List Produced with PROC FORMS

```
Gabrielli, Theresa
24 Ridgetop Rd.
Westboro      MA 01581

Clayton, Aria
314 Bridge St.
Hanover       NH 03755

Dix, Martin L.
4 Shepherd St.
Norwich       VT 05055

Slater, Emily C.
2009 Cherry St.
York          PA 17407

Ericson, Jane
211 Clancey Court
Chapel Hill   NC 27514

An, Ing
95 Willow Dr.
Charlotte     NC 28211
```

Output 1.2 shows a customized version of the same mailing list. The statements that create this list

- invert the name so the first name appears first
- eliminate extra spaces between the city and state
- place three form units in each row
- make three copies of each form
- use only observations from the states in New England.

For an explanation of the program that produces these labels, see Example 3 on page 16.

Output 1.2 Customized Mailing List Produced with PROC FORMS

```
Theresa Gabrielli      Theresa Gabrielli      Theresa Gabrielli
24 Ridgetop Rd.      24 Ridgetop Rd.      24 Ridgetop Rd.
Westboro MA 01581    Westboro MA 01581    Westboro MA 01581

Aria Clayton         Aria Clayton         Aria Clayton
314 Bridge St.       314 Bridge St.       314 Bridge St.
Hanover NH 03755     Hanover NH 03755     Hanover NH 03755

Martin L. Dix         Martin L. Dix         Martin L. Dix
4 Shepherd St.       4 Shepherd St.       4 Shepherd St.
Norwich VT 05055     Norwich VT 05055     Norwich VT 05055
```

Syntax: FORMS Procedure

Requirements: At least one LINE statement

Reminder: You can use the ATTRIB, FORMAT, and WHERE statements. You can also use any global statements. See the “Fundamental Concepts for Using Base SAS Procedures” chapter in *Base SAS Procedures Guide* for a list.

```
PROC FORMS <option(s)>;
  BY <DESCENDING> variable-1
    <...<DESCENDING> variable-n>
    <NOTSORTED>;
  FREQ variable;
  LINE line-number variable(s) </ option(s)>;
```

To do this	Use this statement
Produce a separate set of forms for each BY group	BY
Treat observations as if they appear multiple times in the input data set	FREQ
Specify the information to print on a line of the form unit	LINE

PROC FORMS Statement

```
PROC FORMS <option(s)>;
```

To do this	Use this option
Specify the input data set	DATA=
Identify an external file for PROC FORMS to write to	FILE=
Control the dimensions of a form	
Specify the number of lines in the form unit	LINES=
Specify the number of columns across the form unit	WIDTH=
Control the placement of the forms	
Specify the number of form units to print across the page	ACROSS=
Specify the number of spaces to print between form units	BETWEEN=
Specify the number of lines to skip on a page before printing the first form unit	DOWN=

To do this	Use this option
Specify the number of spaces to indent before printing the first form unit in each row	INDENT=
Specify the number of form units to print down the page	NDOWN=
Specify the number of lines on a page of forms	PAGESIZE=
Specify the number of lines to skip between form units	SKIP=
Control the number of each form unit that PROC FORMS prints	
Specify the number of form units to produce for each observation in each set of form units	COPIES=
Specify the number of sets of form units to produce	SETS=
Control the placement of page-eject characters	CC
Specify the number of lines of dummy form units to print	ALIGN=

Options

ACROSS=*form-units-per-line*

specifies the number of form units to print across the page. (See Figure 1.1 on page 10.)

Alias: A=

Default: 1

Range: 1–200

Featured in: Example 1 on page 11

ALIGN=*number*

controls the number of alignment form units that print before the actual data values. Use the alignment form units, which consist solely of Xs, to check printer alignment.

Default: 8 with FILE=; 0 without FILE=

Interaction: If you use ACROSS=, then the number of dummy form units that print is the product of the values of ACROSS= and ALIGN=.

Featured in: Example 1 on page 11

BETWEEN=*spaces-between-form-units*

specifies the number of spaces to print between form units. (See Figure 1.1 on page 10.)

Alias: B=

Default: 1

Range: 1–200

Featured in: Example 1 on page 11

CC

in continuous mode, writes a page-eject character at the top of the first page. In page mode, if you also specify FILE=, then CC writes a page-eject character at the top of each page. (CC has no effect if you omit FILE=.) For a discussion of page mode and continuous mode, see “Modes of Operation” on page 10.

Tip: If you omit CC, then PROC FORMS issues blank lines to go to the next page. It is recommended that you always use CC with page-mode operation.

Featured in: Example 2 on page 14

COPIES=number

specifies the number of form units to produce for each observation in each set of form units. All copies of an observation appear together.

Alias: C=

Default: 1

Featured in: Example 3 on page 16

DATA=SAS-data-set

identifies the input SAS data set.

DOWN=top-margin

specifies the number of lines to skip on a page before printing the first form unit. (See Figure 1.1 on page 10.)

Alias: D=

Default: 1

Range: 1–200

Featured in: Example 1 on page 11

Note: When PROC FORMS writes to the procedure output file, it uses one line for each TITLE statement and leaves a blank line beneath the last title. Counting for the top margin begins at the next line. Thus, if you have two TITLE statements and specify DOWN=5, then PROC FORMS begins printing the first form unit on each page on line 9. △

FILE=fileref

identifies an external file for PROC FORMS to write to. Use the FILENAME statement to associate an external file with a fileref (see *SAS Language Reference: Concepts*).

Alias: DDNAME=, D=

Default: If you omit FILE=, then PROC FORMS writes to the procedure output file and selects page mode.

Interaction: If you use FILE= and do not specify the ALIGN= option, then PROC FORMS uses ALIGN=8.

Interaction: When you use FILE=, PROC FORMS uses the value of DOWN= only on the first page of form units.

Interaction: If you use FILE= with NDOWN= or PAGESIZE= or both, then select page mode. Otherwise, select continuous mode.

Featured in: Example 1 on page 11

INDENT=left-margin

specifies the number of spaces to indent before printing the first form unit in each row. (See Figure 1.1 on page 10.)

Alias: I=

Default: 0

Range: 0–200

LINES=form-unit-length

specifies the number of lines in a form unit. (See Figure 1.1 on page 10.)

Alias: L=

Default: the largest number that is used with the LINE statement

Range: 1–200

NDOWN=form-units-per-page

specifies the number of form units to print down the page and selects page-mode operation. (See Figure 1.1 on page 10.)

Alias: ND=

Default: $\text{FLOOR}((\text{PAGESIZE}-\text{DOWN}+\text{SKIP})/(\text{LINES}+\text{SKIP}))$ where FLOOR is a SAS function that returns the largest integer less than or equal to the value of the argument.

Interaction: If NDOWN= specifies a number of form units that is less than PAGESIZE= allows, then PROC FORMS uses the value of the NDOWN= option. If NDOWN= specifies a number of form units that is greater than PAGESIZE= allows, then PROC FORMS adjusts the value of NDOWN= downward to accommodate the page size.

Featured in: Example 2 on page 14

PAGESIZE=*lines-per-page*

specifies the number of lines on a page of forms after allowing for TITLE statements and a blank line after the titles. (See Figure 1.1 on page 10.) PAGESIZE= also selects page-mode operation.

Alias: PS=

Default: the system page size (with FILE=); inferred from the characteristics of the procedure output file and from title information (without FILE=)

Range: the value of DOWN= plus the value of LINES=, up to 10,000

Interaction: When you write to the procedure output, if the page size that you specify is greater than the page size that is specified by the SAS system option PAGESIZE=, then PROC FORMS adjusts the PROC FORMS page size to accommodate the system page size.

Interaction: If the page size allows for more form units than NDOWN= specifies, then PROC FORMS uses the value of the NDOWN= option. If the page size does not allow for as many form units as NDOWN= specifies, then PROC FORMS adjusts the value of NDOWN= to accommodate the page size.

SETS=*number*

specifies the number of sets of form units to produce. In page-mode operation, PROC FORMS starts each set on a new page.

Default: 1

Featured in: Example 2 on page 14

SKIP=*lines-between-form-units*

specifies the number of lines to skip between form units. (See Figure 1.1 on page 10.)

Alias: S=

Default: 1

Range: 1–200

Featured in: Example 1 on page 11

WIDTH=*form-unit-width*

specifies the number of columns across the form unit. PROC FORMS truncates values that do not fit in the specified width. (See Figure 1.1 on page 10.)

Alias: W=

Default: width of the widest line

Range: 1–255

Featured in: Example 1 on page 11

BY Statement

Produces a separate set of forms for each BY group.

Main discussion: “Statements with the Same Function in Multiple Procedures” chapter in *Base SAS Procedures Guide*

```
BY <DESCENDING> variable-1  
  <...<DESCENDING> variable-n>  
  <NOTSORTED>;
```

Required Arguments

variable

specifies the variable that the procedure uses to form BY groups. You can specify more than one variable. If you do not use the NOTSORTED option in the BY statement, then either the observations in the data set must be sorted by all the variables that you specify, or they must be indexed appropriately. Variables in a BY statement are called *BY variables*.

Options

DESCENDING

specifies that the data set is sorted in descending order by the variable that immediately follows the word DESCENDING in the BY statement.

NOTSORTED

specifies that observations are not necessarily sorted in alphabetic or numeric order. The data is grouped in another way, for example, chronological order.

The requirement for ordering or indexing observations according to the values of BY variables is suspended for BY-group processing when you use the NOTSORTED option. In fact, the procedure does not use an index if you specify NOTSORTED. The procedure defines a BY group as a set of contiguous observations that have the same values for all BY variables. If observations with the same values for the BY variables are not contiguous, then the procedure treats each contiguous set as a separate BY group.

How PROC FORMS Separates BY Groups

In page mode, the forms for each BY group begin on a new page. In continuous mode, BY groups are not separated.

FREQ Statement

Treats observations as if they appear multiple times in the input data set.

```
FREQ variable;
```

Required Arguments

variable

specifies a numeric variable whose value represents the frequency of each observation. If you use the `FREQ` statement, then the procedure assumes that each observation in the input data set represents n observations, where n is the value of *variable*. If n is not an integer, then `PROC FORMS` truncates it. If n is less than 1 (which includes missing), then the procedure does not use that observation.

The sum of the frequency variable represents the total number of observations.

LINE Statement

Specifies the information to print on one line of the form unit. Use one `LINE` statement for each line of the form unit.

LINE *line-number* *variable(s)* *</option(s)>*;

To do this	Use this option
Specify the number of spaces to indent the line within the form unit	<code>INDENT=</code>
Rotate the words in a character variable that contains a comma around the comma and remove the comma	<code>LASTNAME</code>
Remove extra blanks from the line so that one blank separates variables	<code>PACK</code>
Remove periods that represent missing values from a line that contains no other values.	<code>REMOVE</code>

Required Arguments

line-number

identifies the number of the line. You can specify lines in any order. You do not need a `LINE` statement for a blank line.

Range: An integer between 1 and the value of `LINES=` in the `PROC FORMS` statement

variable(s)

specifies one or more variables to print on this line of the form unit. The `FORMS` procedure inserts one space between each value. By default, the width of a variable's field in the form unit is the formatted length of that variable. Default formats are the length of the variable for character variables and `BEST12.` for numeric variables.

Interaction: If the length of all values in a line is greater than the value of `WIDTH=` that is specified in the `PROC FORMS` statement, then `PROC FORMS` truncates

the values (starting with the rightmost value in the line) to fit the WIDTH= value. For information about fitting more variables onto a line, see PACK on page 9.

Options

INDENT=*margin-within-form-unit*

specifies the number of spaces to indent the line within the form unit. Contrast this option to INDENT= in the PROC FORMS statement, which specifies the size of the left margin that precedes the first form unit in each row.

Alias: I=

Featured in: Example 1 on page 11

LASTNAME

rotates the words in a character variable that contains a comma around the comma and removes the comma.

Alias: L

Featured in: Example 1 on page 11

PACK

removes extra blanks from the line so that one blank separates variables.

Alias: P

Tip: PACK can fit more fields onto a form unit, but if the values for all the variables are long, then you might lose an entire field. To avoid this problem, use a FORMAT statement to limit the space for each variable. For example, the following statement sets the field widths of the variables CITY and STATE to 20 and 2 columns, respectively:

```
format city $20. state $2.;
```

Featured in: Example 1 on page 11

REMOVE

removes periods that represent missing values from a line that contains no other values.


Alias: R

Concepts: FORMS Procedure

Form Layout

The size and spacing of form units are controlled by options in the PROC FORMS statement, as illustrated in Figure 1.1 on page 10. (See also the discussion of these options on page 4.)

Figure 1.1 Sample Placement for Forms



The values of the variables that are specified in `LINE` statements are formatted into a form unit that is `WIDTH=` columns wide and `LINES=` lines long. Values that do not fit into `WIDTH=` columns are truncated. `ACROSS=` form units are printed across the page, with `BETWEEN=` spaces between adjacent form units. The forms are indented `INDENT=` spaces from the left margin. `SKIP=` blank lines are printed between form units down the page.

Modes of Operation

PROC FORMS Operation Modes

PROC FORMS operates in two modes: continuous mode and page mode. Continuous mode is for forms that feed continuously through a printer, without the printer's needing to perform page ejects. Page mode is for forms that use separate sheets of paper for each form unit or for multiple form units (such as sheets of labels that come with 30 labels per sheet of paper).

By default, PROC FORMS uses page mode. To select continuous mode, you must specify `FILE=` and must not specify `NODOWN=` or `PAGESIZE=`.

In Continuous Mode, PROC FORMS Always Writes to an External File

When it writes in continuous mode, PROC FORMS

- 1 skips the number of lines that are specified by `DOWN=`
- 2 prints one form unit
- 3 skips the number of lines that are specified by `SKIP=`
- 4 repeats steps 2 and 3 until it uses all the data.

By default, in continuous mode the first eight form units are dummy form units that consist solely of Xs. These forms give the printer operator a chance to align the printer before real form units begin to print. Use `ALIGN=` to alter the number of dummy form units. Once the dummy form units are aligned to the physical forms, the file prints correctly. Carriage control characters are unnecessary.

In Page Mode, PROC FORMS Can Write Either to an External File or to the Procedure Output File

In page mode, PROC FORMS

- 1 goes to the top of a new page
- 2 skips the number of lines that are specified by DOWN=
- 3 prints the number of form units that are specified by NDOWN= down the page, or if you omit NDOWN=, prints the maximum number of form units that are allowed by the page size
- 4 repeats steps 1 through 3 until it uses all the data.

When PROC FORMS has written as many form units as you specify, either it writes a blank line for each line that remains on the page (as determined by the PAGESIZE= option) or it writes a page-eject character. If you are writing to the procedure output file, then PROC FORMS always writes the page-eject characters. If you have specified FILE=, then PROC FORMS by default writes blank lines, but if you specify the CC option, then it writes page-eject characters instead.

In page mode, the easiest way to ensure proper alignment is to specify the number of form units to print down the page with the NDOWN= option and to use CC to write a page-eject character at the beginning of each page. If you omit CC, then be sure that the page size is set correctly. If the page size is not set correctly, then PROC FORMS will not write the number of blank lines that are needed to arrive at the top of the next page.

Note: It is recommended that you always use CC when you use page mode with the FILE= option. Δ

Note: The procedure output file contains some things that you might not want on your forms. If you omit the FILE= option, then the output from PROC FORMS goes to the procedure output file. If the DATE and NUMBER options are in effect, then the output will contain dates and page numbers. If any titles or footnotes are defined, then they will appear in the output as well. Δ

Examples: FORMS Procedure

The examples in this chapter assume alignment for the forms that they use. You must experiment to determine how to align your form units with your forms.

Example 1: Printing a Single Form Unit for Each Observation

Procedure features:

PROC FORMS statement options:

ACROSS=
ALIGN=
BETWEEN=
DOWN=
FILE=
SKIP=
WIDTH=

LINE statement options:

INDENT=
LASTNAME
PACK

Other features:

SORT procedure

This example uses PROC FORMS to print one set of mailing labels that consists of one copy of the form unit for each observation.

Program

Create the LIST data set and sort by zip code. The data set LIST contains names and mailing addresses. PROC SORT sorts the data by zip code.

```
options nodate pageno=1 linesize=80 pagesize=60;
data list;
  input Name $ 1-19 Street $ 20-39 City $ 40-54
        State $ 55-56 Zip $ 59-63;
  datalines;
Ericson, Jane      211 Clancey Court   Chapel Hill   NC  27514
Dix, Martin L.    4 Shepherd St.     Norwich      VT  05055
Gabrielli, Theresa 24 Ridgetop Rd.    Westboro     MA  01581
Clayton, Aria     314 Bridge St.     Hanover      NH  03755
Archuleta, Ruby   Box 108             Milagro      NM  87429
Misiewicz, Jeremy 43-C Lakeview Apts. Madison      WI  53704
Ahmadi, Hafez     5203 Marston Way   Boulder      CO  80302
Jacobson, Becky   7 Lincoln St.      Tallahassee FL  32312
An, Ing           95 Willow Dr.      Charlotte    NC  28211
Slater, Emily C.  2009 Cherry St.    York         PA  17407
;

proc sort data=list;
  by zip;
run;
```

Specify a name for the external file. The FILENAME statement associates the name LABELS with the external file that will receive the output from PROC FORMS.

```
filename labels 'external-file';
```

Send a single form unit for each observation to the LABELS external file. FILE= sends the output to the file that is associated with the fileref LABELS. Because neither NDOWN= nor PAGESIZE= is specified, PROC FORMS uses continuous mode. WIDTH= sets the width of the form units to 24 in order to provide enough space for all the variables on each line. ACROSS= writes three form units across each page. BETWEEN= inserts four blank characters between adjacent form units. DOWN= skips two lines at the top of the file so that the form units and the forms align correctly. SKIP= skips two lines between form units to maintain the proper alignment. ALIGN= prints two lines of dummy form units.

```
proc forms data=list file=labels
  width=24
  across=3
  between=4
```



```
down=2
skip=2
align=2;
```

Specify the variables to place on each line. The LINE statements specify the variables to place on each line. LASTNAME removes the comma from Name and writes the first name before the last name. PACK removes extra blank characters between City and State. INDENT= indents Zip 15 spaces.

```
line 1 name / lastname;
line 2 street;
line 3 city state / pack;
line 4 zip / indent=15;
run;
```

Output

XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX	XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX	XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX
XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX	XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX	XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX XXXXXXXXXXXXXXXXXXXXX
Theresa Gabrielli 24 Ridgetop Rd. Westboro MA 01581	Aria Clayton 314 Bridge St. Hanover NH 03755	Martin L. Dix 4 Shepherd St. Norwich VT 05055
Emily C. Slater 2009 Cherry St. York PA 17407	Jane Ericson 211 Clancey Court Chapel Hill NC 27514	Ing An 95 Willow Dr. Charlotte NC 28211
Becky Jacobson 7 Lincoln St. Tallahassee FL 32312	Jeremy Misiewicz 43-C Lakeview Apts. Madison WI 53704	Hafez Ahmadi 5203 Marston Way Boulder CO 80302
Ruby Archuleta Box 108 Milagro NM 87429		

Example 2: Printing Two Sets of Mailing Labels

Procedure features:

PROC FORMS statement options:

ALIGN=

CC

FILE=

NDOWN=

SETS=

Data set:

LIST on page 12

This example uses page mode and SETS= to produce two sets of mailing labels. Each sheet of labels holds four rows of two labels.

Program

Specify a name for the external file. The FILENAME statement associates the name LABELS with the external file that will receive the output from PROC FORMS.

```
filename labels 'external-file';
```

```
options nodate pageno=1 linesize=80 pagesize=60 ;
```

Send two sets of form units to the LABELS external file. FILE= sends the output to the file that is associated with the fileref LABELS. NDOWN= prints four rows of form units on each page. CC writes carriage control characters to the file that is specified by FILE=. WIDTH= sets the width of the form units to 24 to provide enough space for the variables on each line. ACROSS= writes two form units across each page. BETWEEN= inserts 20 blank characters between adjacent form units. DOWN= skips two lines at the top of each page so that the form units and the forms align correctly. SKIP= skips three lines between form units to maintain the proper alignment. ALIGN= suppresses the printing of dummy form units. SETS= writes two sets of form units. Each set begins on a new page.

```
proc forms data=list file=labels
  ndown=4
  cc
  width=24
  across=2
  between=20
  down=2
  skip=3
  align=0
  sets=2;
```

Specify the variables to place on each line. The LINE statements specify the variables to place on each line. PACK removes extra blank characters between City and State.

```

line 1 name;
line 2 street;
line 3 city state zip / pack;
run;

```

Output

Gabrielli, Theresa
24 Ridgetop Rd.
Westboro MA 01581

Clayton, Aria
314 Bridge St.
Hanover NH 03755

Dix, Martin L.
4 Shepherd St.
Norwich VT 05055

Slater, Emily C.
2009 Cherry St.
York PA 17407

Ericson, Jane
211 Clancey Court
Chapel Hill NC 27514

An, Ing
95 Willow Dr.
Charlotte NC 28211

Jacobson, Becky
7 Lincoln St.
Tallahassee FL 32312

Misiewicz, Jeremy
43-C Lakeview Apts.
Madison WI 53704

Ahmadi, Hafez
5203 Marston Way
Boulder CO 80302

Archuleta, Ruby
Box 108
Milagro NM 87429

Gabrielli, Theresa
24 Ridgetop Rd.
Westboro MA 01581

Clayton, Aria
314 Bridge St.
Hanover NH 03755

Dix, Martin L.
4 Shepherd St.
Norwich VT 05055

Slater, Emily C.
2009 Cherry St.
York PA 17407

Ericson, Jane
211 Clancey Court
Chapel Hill NC 27514

An, Ing
95 Willow Dr.
Charlotte NC 28211

Jacobson, Becky
7 Lincoln St.
Tallahassee FL 32312

Misiewicz, Jeremy
43-C Lakeview Apts.
Madison WI 53704

```
Ahmadi, Hafez
5203 Marston Way
Boulder CO 80302
```

```
Archuleta, Ruby
Box 108
Milagro NM 87429
```

Example 3: Writing Multiple Copies of a Label within a Single Set of Labels

Procedure features:

PROC FORMS statement options:

COPIES=

LINE statement options:

LASTNAME

PACK

Data set: LIST on page 12

This example writes one set of mailing labels that consists of three copies of each form unit. It selects only those observations with addresses in one of the New England states.

Program

Specify a name for the external file. The FILENAME statement associates the name LABELS with the external file that will receive the output from PROC FORMS.

```
filename labels 'external-file';
```

```
options pagesize=60 pageno=1 nodate linesize=80;
```

Send three copies of each form unit to the LABELS external file. FILE= sends the output to the file that is associated with the fileref LABELS. NDOWN= prints five rows of form units on each page. CC writes carriage control characters to the file that is specified by FILE=. ALIGN= suppresses the printing of dummy form units. WIDTH= sets the width of the form units to 24 to provide enough space for the variables on each line. ACROSS= writes three form units across each page. DOWN= skips two lines at the top of each page so that the form units and the forms align correctly. SKIP= skips two lines between form units to maintain the proper alignment. COPIES= writes three copies of each form unit.

```
proc forms data=list file=labels
      ndown=5
      cc
      align=0
      width=24
      across=3
      down=2
      skip=2
```

```
copies=3;
```

Specify the variables to place on each line. The LINE statements specify the variables to place on each line. LASTNAME removes the comma from Name and writes the first name before the last name. PACK removes extra blank characters between City and State.

```
line 1 name / lastname;
line 2 street;
line 3 city state zip / pack;
```

Specify the observations for the external file. The WHERE statement selects observations in which State is one of the New England states.

```
where state in('ME', 'NH', 'VT', 'MA', 'CT', 'RI');
run;
```

Output

Theresa Gabrielli 24 Ridgetop Rd. Westboro MA 01581	Theresa Gabrielli 24 Ridgetop Rd. Westboro MA 01581	Theresa Gabrielli 24 Ridgetop Rd. Westboro MA 01581
Aria Clayton 314 Bridge St. Hanover NH 03755	Aria Clayton 314 Bridge St. Hanover NH 03755	Aria Clayton 314 Bridge St. Hanover NH 03755
Martin L. Dix 4 Shepherd St. Norwich VT 05055	Martin L. Dix 4 Shepherd St. Norwich VT 05055	Martin L. Dix 4 Shepherd St. Norwich VT 05055

Your Turn

If you have comments or suggestions about *Proc FORMS*, please send them to us on a photocopy of this page, or send us electronic mail.

For comments about this book, please return the photocopy to

SAS Publishing
SAS Campus Drive
Cary, NC 27513
email: yourturn@sas.com

For suggestions about the software, please return the photocopy to

SAS Institute Inc.
Technical Support Division
SAS Campus Drive
Cary, NC 27513
email: suggest@sas.com